

HP 15C Calculator Programs

Dr. D.G. Simpson
Department of Physical Sciences and Engineering
Prince George's Community College

May 3, 2014

These calculator programs are written for the Hewlett-Packard HP 15C scientific calculator, and can be easily modified to run on other HP calculators that use HP RPN.

Contents

- 1. Projectile Problem**
- 2. Kepler's Equation**
- 3. Hyperbolic Kepler's Equation**
- 4. Barker's Equation**
- 5. Reduction of an Angle**
- 6. Helmert's Equation**
- 7. Pendulum Period**
- 8. 1D Perfectly Elastic Collisions**

1 Projectile Problem

This program solves the following projectile problem: given a target sitting on a hill at coordinates (x_t, y_t) and a cannon at the origin with muzzle velocity v_0 , at what angle θ should the cannon be aimed to hit the target? The solution is found numerically using Newton's method.

To run the program, enter:

v_0 ENTER x_t ENTER y_t ENTER θ_0 f A

Here v_0 , x_t , and y_t may be in any consistent set of units, and the angle θ_0 (the initial estimate of the answer) is in degrees. The program returns the angle θ needed to hit the target in degrees.

After running the program, the calculator will be set to degrees mode.

Program Listing

```
001- 42,21,11    f LBL A
002-   43  8     g RAD
003-   42  3     f →RAD
004-   44  .3    STO .3
005-           33  R↓
006-   44  .1    STO .1
007-           33  R↓
008-   44  .0    STO .0
009-           33  R↓
010-   44  .2    STO .2
011-           1   1
012-           48  .
013-           0   0
014-           1   1
015-           4   4
016-   44  .4    STO .4
017-   45  .3    RCL .3
018- 42,21, 0   f LBL 0
019-           2   2
020-           20  ×
021-           23  SIN
022-   45  .0    RCL .0
023-           20  ×
024-   45  .3    RCL .3
025-           24  COS
026-   43  11    g  $x^2$ 
027-   45  .1    RCL .1
028-           20  ×
029-           2   2
```

```

030-      20      ×
031-      30      -
032-     45 .0    RCL .0
033-     45 .2    RCL .2
034-      10      ÷
035-     43 11    g x2
036-      9       9
037-      48      .
038-      8       8
039-      20      ×
040-      30      -
041-     45 .3    RCL .3
042-      2       2
043-      20      ×
044-      24      COS
045-     45 .0    RCL .0
046-      20      ×
047-      2       2
048-      20      ×
049-     45 .3    RCL .3
050-      2       2
051-      20      ×
052-      23      SIN
053-     45 .1    RCL .1
054-      20      ×
055-      2       2
056-      20      ×
057-      40      +
058-      10      ÷
059-     45 .3    RCL .3
060-      34      x ≥ y
061-      30      -
062-     44 .3    STO .3
063- 42, 6, .4   f ISG .4
064-      22 0    GTO 0
065-     43 3    g →DEG
066-     43 7    g DEG
067-     43 32   g RTN

```

Example. Let $v_0 = 30$ m/s, $(x_t, y_t) = (50 \text{ m}, 20 \text{ m})$, and $\theta_0 = 30^\circ$. Enter the above program, then type:

```
30 ENTER 50 ENTER 20 ENTER 30 f A
```

The program returns $\theta = 41.5357^\circ$.

2 Kepler's Equation

Given the mean anomaly M (in degrees) and the orbit eccentricity e , this program solves Kepler's equation

$$M = E - e \sin E$$

to find the eccentric anomaly E . This is a very simple implementation—it includes no convergence test, and simply solves Kepler's equation by performing 15 iterations of Newton's method.

To run the program, enter:

M ENTER e f A

where M is in degrees. The program returns the eccentric anomaly E in degrees. After running the program, the calculator will be set to degrees mode.

Program Listing

```
001- 42,21,11    f LBL A
002-    44 .0    STO .0
003-    34      x ≥ y
004-    42 3     f →RAD
005-    44 .1    STO .1
006-    44 .2    STO .2
007-    43 8     g RAD
008-    1        1
009-    48      .
010-    0        0
011-    1        1
012-    4        4
013-    44 .3    STO .3
014-    45 .2    RCL .2
015- 42,21, 0   f LBL 0
016-    45 .1    RCL .1
017-    45 .2    RCL .2
018-    30      -
019-    45 .2    RCL .2
020-    23      SIN
021-    45 .0    RCL .0
022-    20      ×
023-    40      +
024-    45 .2    RCL .2
025-    24      COS
026-    45 .0    RCL .0
027-    20      ×
```

```

028-      1      1
029-      30     -
030-      10     ÷
031-      30     -
032-     44 .2   STO .2
033-  42, 6, .3  f ISG .3
034-     22  0   GTO 0
035-     43  3   g →DEG
036-     43  7   g DEG
037-     43 32   g RTN

```

Example. Let $M = 60^\circ$, $e = 0.15$. Enter the above program, then type:

```
60  ENTER  .15  f A
```

The program returns $E = 67.9667^\circ$.

3 Hyperbolic Kepler's Equation

Given the mean anomaly M (in degrees) and the orbit eccentricity e , this program solves the hyperbolic Kepler equation

$$M = e \sinh F - F$$

to find the variable F . This is a very simple implementation—it includes no convergence test, and simply solves the hyperbolic Kepler equation by performing 15 iterations of Newton's method.

To run the program, enter:

M ENTER e f A

where M is in degrees. The program returns the variable F .

Program Listing

```
001- 42,21,11    f LBL A
002-    44 .0    STO .0
003-    34      x ≥ y
004-    42 3    f →RAD
005-    44 .1    STO .1
006-    44 .2    STO .2
007-    1      1
008-    48      .
009-    0      0
010-    1      1
011-    4      4
012-    44 .3    STO .3
013-    45 .2    RCL .2
014- 42,21, 0    f LBL 0
015-    45 .1    RCL .1
016-    45 .2    RCL .2
017-    40      +
018-    45 .2    RCL .2
019- 42,22,23    f HYP SIN
020-    45 .0    RCL .0
021-    20      ×
022-    30      -
023-    45 .2    RCL .2
024- 42,22,24    f HYP COS
025-    45 .0    RCL .0
026-    20      ×
027-    1      1
028-    34      x ≥ y
```

1

```

029-      30      -
030-      10      ÷
031-      30      -
032-     44 .2     STO .2
033-  42, 6, .3   f ISG .3
034-      22  0   GTO 0
035-      43 32   g RTN

```

Example. Let $M = 60^\circ$, $e = 1.15$. Enter the above program, then type:

```
60  ENTER  1.15  f A
```

The program returns $F = 1.5555$.

4 Barker's Equation

Given the constant $K = \sqrt{GM/(2q^3)}(t - T_p)$, this program solves Barker's equation

$$\tan\left(\frac{f}{2}\right) + \frac{1}{3}\tan^3\left(\frac{f}{2}\right) = \sqrt{\frac{GM}{2q^3}}(t - T_p)$$

to find the true anomaly f .

To run the program, enter the dimensionless number

$$K = \sqrt{\frac{GM}{2q^3}}(t - T_p)$$

as follows:

K f A

The program returns the anomaly f .

The program will work in either Degrees or Radians mode.

Program Listing

```
001- 42,21,11    f LBL A
002-    44 .0    STO .0
003-    43 16    g ABS
004-       1    1
005-       48    .
006-       5    5
007-       20    ×
008-       36    ENTER
009-       36    ENTER
010-       20    ×
011-       1    1
012-       40    +
013-       11    √x
014-       40    +
015-       3    3
016-       15    1/x
017-       14    yx
018-       36    ENTER
019-       36    ENTER
020-       20    ×
021-       1    1
022-       30    -
023-       34    x ≥ y
024-       10    ÷
```

```

025-    45 .0    RCL .0
026-        36    ENTER
027-    43 16    g ABS
028-        10    ÷
029-        20    ×
030-    43 25    g TAN-1
031-        2     2
032-        20    ×
033-    43 32    g RTN

```

Example. Let $K = 19.38$ and set the calculator's angle mode to degrees. Enter the above program, then type:

```
19.38    f A
```

The program returns $f = 149.0847^\circ$.

5 Reduction of an Angle

This program reduces a given angle to the range $[0, 360^\circ)$ in degrees mode, or $[0, 2\pi)$ in radians mode. It will work correctly whether the calculator is set for degrees or radians mode.

To run the program:

θ f A

The program will return the equivalent reduced angle.

Program Listing

```
001- 42,21,11    f LBL A
002-    44 .1    STO .1
003-    1       1
004-    16       CHS
005-    43 24    COS-1
006-    2       2
007-    20       ×
008-    44 .0    STO .0
009-    45 .1    RCL .1
010- 43,30, 3    g TEST 3
011-    22 0     GTO 0
012-    45 .0    RCL .0
013-    10       ÷
014-    16       CHS
015-    43 44    g INT
016-    1       1
017-    40       +
018-    45 .0    RCL .0
019-    20       ×
020-    45 .1    RCL .1
021-    40       +
022-    43 32    g RTN
023- 42,21, 0    f LBL 0
024-    45 .0    RCL .0
025-    34        $x \geq y$ 
026- 43,30, 8    g TEST 8
027-    43 32    g RTN
028-    34        $x \geq y$ 
029-    10       ÷
030-    43 44    g INT
031-    45 .0    RCL .0
032-    20       ×
033-    45 .1    RCL .1
```

```
034-      34      x ≥ y
035-      30      -
036-    43 32      g RTN
```

Example. Let $\theta = 5000^\circ$ and set the calculator's angle mode to degrees. Enter the above program, then type:

```
5000  f A
```

The program returns 320° .

6 Helmert's Equation

Given the latitude θ (in degrees) and the elevation H (in meters), this program uses Helmert's equation to find the acceleration due to gravity g .

To run the program, enter:

θ ENTER H f A

where θ is in degrees and H is in meters. The program returns the acceleration due to gravity g in m/s^2 .

After running the program, the calculator will be set to degrees mode.

Program Listing

```
001- 42,21,11    f LBL A
002-   43  7     g DEG
003-   34       x  $\geq$  y
004-   2        2
005-   20        $\times$ 
006-   44 .0    STO .0
007-   24      COS
008-   45 .1    RCL .1
009-   20        $\times$ 
010-   45 .2    RCL .2
011-   34       x  $\geq$  y
012-   30      -
013-   45 .0    RCL .0
014-   24      COS
015-   43 11    g  $x^2$ 
016-   45 .3    RCL .3
017-   20        $\times$ 
018-   40      +
019-   34       x  $\geq$  y
020-   45 .4    RCL .4
021-   20        $\times$ 
022-   30      -
023-   43 32    g RTN
```

R.1 = 0.025928

R.2 = 9.80616

R.3 = 6.9×10^{-5}

R.4 = 3.086×10^{-6}

Example. Let $\theta = 38.898^\circ$, $H = 53$ m. Enter the above program, then type:

38.898 ENTER 53 f A

The program returns $g = 9.80052$ m/s².

7 Pendulum Period

Given the length L and amplitude θ of a simple plane pendulum, this program finds the exact period T , using the arithmetic-geometric mean method.

To run the program, enter:

L ENTER θ f A

where L is in meters and θ is in degrees. The program returns the period T in seconds.

After running the program, the calculator will be set to degrees mode.

Program Listing

```
001- 42,21,11    f LBL A
002-   43  7     g DEG
003-   44  .0    STO .0
004-   44  34    x ≥ y
005-   44  .1    STO .1
006-   44  1     1
007-   45  .0    RCL .0
008-   44  2     2
009-   44  10    ÷
010-   44  24    COS
011-   44  40    +
012-   44  2     2
013-   44  10    ÷
014-   44  .2    STO .2
015-   45  .0    RCL .0
016-   44  2     2
017-   44  10    ÷
018-   44  24    COS
019-   44  11    √x
020-   44  .3    STO .3
021-   44  1     1
022-   44  48    .
023-   44  0     0
024-   44  1     1
025-   44  .4    STO .4
026- 42,21, 0    f LBL 0
027-   45  .2    RCL .2
028-   44  36    ENTER
029-   44  36    ENTER
030-   45  .3    RCL .3
031-   44  40    +
032-   44  2     2
```

```

033-      10      ÷
034-     44 .2    STO .2
035-      33      R↓
036-     45 .3    RCL .3
037-      20      ×
038-      11      √x
039-     44 .3    STO .3
040- 42, 6, .4   f ISG .4
041-     22 0     GTO 0
042-     45 .1    RCL .1
043-      9       .
044-     48      .
045-      8       8
046-     10      ÷
047-     11      √x
048-      2       2
049-     20      ×
050-     43 26   g π
051-     20      ×
052-     45 .2    RCL .2
053-     10      ÷
054-     43 32   g RTN

```

Example. Let $L = 1.2$ m and $\theta = 65^\circ$. Enter the above program, then type:

```
1.2  ENTER  65  f A
```

The program returns $T = 2.3898$ sec.

8 1D Perfectly Elastic Collisions

Given the masses m_1 and m_2 of two bodies and their initial velocities v_{1i} and v_{2i} , this program finds the post-collision velocities v_{1f} and v_{2f} using

$$v_{1f} = \left(\frac{m_1 - m_2}{m_1 + m_2} \right) v_{1i} + \left(\frac{2m_2}{m_1 + m_2} \right) v_{2i}$$
$$v_{2f} = \left(\frac{2m_1}{m_1 + m_2} \right) v_{1i} + \left(\frac{m_2 - m_1}{m_1 + m_2} \right) v_{2i}$$

To run the program, enter:

m_1 ENTER m_2 ENTER v_{1i} ENTER v_{2i} f A

The program will return the post-collision velocities v_{1f} (in the X register) and v_{2f} (in the Y register), in the same units.

Program Listing

```
001- 42,21,11    f LBL A
002- 44 .5      STO .5
003- 33        R↓
004- 44 .4      STO .4
005- 33        R↓
006- 44 .2      STO .2
007- 33        R↓
008- 44 .1      STO .1
009- 45 .2      RCL .2
010- 30        -
011- 45 .1      RCL .1
012- 45 .2      RCL .2
013- 40        +
014- 44 .0      STO .0
015- 10        ÷
016- 45 .4      RCL .4
017- 20        ×
018- 2         2
019- 45 .2      RCL .2
020- 20        ×
021- 45 .5      RCL .5
022- 20        ×
023- 45 .0      RCL .0
024- 10        ÷
025- 40        +
026- 44 .3      STO .3
027- 2         2
```

```

028-    45 .1    RCL .1
029-        20    ×
030-    45 .4    RCL .4
031-        20    ×
032-    45 .0    RCL .0
033-        10    ÷
034-    45 .2    RCL .2
035-    45 .1    RCL .1
036-        30    −
037-    45 .0    RCL .0
038-        10    ÷
039-    45 .5    RCL .5
040-        20    ×
041-        40    +
042-    45 .3    RCL .3
043-    43 32    g RTN

```

Example. Let $m_1 = 2.0$ kg, $m_2 = 7.0$ kg, $v_{1i} = 4.0$ m/s, and $v_{2i} = -5.0$ kg. Enter the above program, then type:

```
2 ENTER 7 ENTER 4 ENTER 5 CHS f A
```

The program returns $v_{1f} = -10$ m/s in the X register, and $v_{2f} = -1$ m/s in the Y register.